A typical series solution

A typical series solution

We need to set up a series to workl with. This is done with the Sum command, which is similar to the sigma notation from mathematics.Technically, a series is a special type of data structure in Mathematica. This type is differnet from the expressions we usually use, and so must specifially be indicated to Mathematica.

```
y = Sum[c[i] x^i, {i,0,6}] + O[x]^7
```

$c[0]+c[1]\,x+c[2]\,x^2+c[3]\,x^3+c[4]\,x^4+c[5]\,x^5+c[6]\,x^6+O[x]^7$

OutputForm[*expr*]

prints as a two-dimensional representation of *expr* using only keyboard characters.

```
OutputForm[y]//Normal
```

```
                     2        3        4        5        6
c[0] + x c[1] + x  c[2] + x  c[3] + x  c[4] + x  c[5] + x  c[6]
```

Now using this series, let's work the following typical problem

$$(x^2 - 1)\, y'' + x\, y' - y = 0$$

Insert the series into the differential equation. Label the result of the substituion as de1.

```
de1 = (x^2 -1) D[y, {x,2}] + x D[y,x] - y == 0
```

$(-c[0]-2c[2])-6c[3]x+(3c[2]-12c[4])x^2+(8c[3]-20c[5])x^3+(15c[4]-30c[6])x^4+O[x]^5==0$

```
OutputForm[de1]
```

```
(-c[0] - 2 c[2]) - 6 x c[3] + (3 c[2] - 12 c[4]) x^2 + (8 c[3] - 20 c[5]) x^3
```

Noticew that all of the terms have been gathered autiomatically.

Now we set all coefficients equal to zero. Mathematica has a general purpose commend which will do this automatically, all at once!

```
coeffeqns = LogicalExpand[ de1]
```

$-c[0]-2c[2]==0\ \&\&\ -6c[3]==0\ \&\&\ 3c[2]-12c[4]==0\ \&\&\ 8c[3]-20c[5]==0\ \&\&\ 15c[4]-30c[6]==0$

Notice all of the double ampersands &&. This says the conditions are

-c[0] - 2c[2] == 0

AND

-6 c[3] == 0

AND

3 c[2] - 12 c[4] == 0

AND

8 c[3] - 20 c[5] == 0

AND

15 c[4] - 30 c[6] == 0

It's just like writing all of the conditions on the board in a row.

Now we want to solve for all the coefficients in terms of the first 2 coefficents c[0] and c[1]. We do this with a variant of the Solve command. We order Mathematica to solve for coefficeints c[2], c[3], c[4], ...c[12]. This will treat c[0] and c[1] as constants and solve for everything in terms of them. Rather than exhaustively write out all of c[2], c[3], c[4], c[12] in the Solve command, we use the Table command to automatically generate the table of coefficients we want to solve for. Neat, isn't it ?!

```
coeffs = Solve[ coeffeqns, Table[ c[i], {i,2,6}]]
```

$$\left\{\left\{c[2]\to -\frac{c[0]}{2},\ c[3]\to 0,\ c[4]\to -\frac{c[0]}{8},\ c[5]\to 0,\ c[6]\to -\frac{c[0]}{16}\right\}\right\}$$

Substitute all of these coefficients back into the series.

```
y = y /. coeffs
```

$$\left\{c[0]+c[1]\,x-\frac{1}{2}c[0]\,x^2-\frac{1}{8}c[0]\,x^4-\frac{1}{16}c[0]\,x^6+O[x]^7\right\}$$

```
OutputForm[y]
```

```
                          2          4          6
                         x  c[0]    x  c[0]    x  c[0]       7
{c[0] + c[1] x - ------- - ------- - ------- + O[x] }
                            2          8          16
```

Now we want to pick off the two linearly independent solutions by factoring out the corresponding coefficients. There are many ways to do this, I choose the following way:

Coefficient[*expr*, *form*, *n*]

gives the coefficient of *form*^*n* in *expr*.

```
y1 = Coefficient[ y,c[1]]
```

$\{x\}$

Here's the second solution,   at least the first 7 terms of the series,, out to order 12. It looks like a series, but I don't recognize it as being familiar..

```
y2 = Coefficient[ y, c[0]]
```

$$\left\{1-\frac{x^2}{2}-\frac{x^4}{8}-\frac{x^6}{16}\right\}$$

A Further Investigation of the Solution

Another way to get the second solution is to use the first solution by using the reduction of order technique.

```
y1 = x
```

x

```
 p = x/(x^2 -1)
```

$$\frac{x}{-1+x^2}$$

```
y2 = y1 Integrate[ Exp[ -Integrate[p, x]]/y1^2, x]
```

$$\sqrt{-1+x^2}$$

That's the result of the order of reduction formula for the second solution, starting from the first fairly simple solution. Let's test it to see if it really works in the differential equation.

```
(x^2 -1) D[y2, {x,2}] + x D[y2,x] - y2
```

$$\frac{x^2}{\sqrt{-1+x^2}} - \sqrt{-1+x^2} + \left(-1+x^2\right)\left(-\frac{x^2}{\left(-1+x^2\right)^{3/2}} + \frac{1}{\sqrt{-1+x^2}}\right)$$

```
Simplify[%]
```

```
0
```

Yes it works!!

Now the question remains: how does this simple solution in closed form relate to the solution in series solution obtained in the previous section? To answer this, we take the Taylor series expansion ofht eclosed form solution and compare to the series solution.

Use the following command for automatic Taylor series expansions.

```
Series[y2, {x,0,6}]
```

$$i - \frac{i\,x^2}{2} - \frac{i\,x^4}{8} - \frac{i\,x^6}{16} + O[x]^7$$

```
OutputForm[%]
```

```
    I  2   I  4   I   6       7
I - - x  - - x  - -- x  + O[x]
    2      8      16
```

Something's wrong!! I'm getting lots of imaginary constants I where I didn't expect them!!

If we go back, and examine the solution, we begin to see why. The solution y2 = Sqrt[-1 + x^2]

is NOT defined for x = 0. (or more precisely, it is imaginary at x =

0). With a little thought, we see the solution defined in the neighborhood of x = 0 should be y2 = Sqrt[1 - x^2]. Try it in the equation to make sure it works.

```
y2 = Sqrt[1 - x^2]
```

$$\sqrt{1-x^2}$$

```
(x^2 -1) D[y2, {x,2}] + x D[y2,x] - y2
```

$$-\frac{x^2}{\sqrt{1-x^2}}-\sqrt{1-x^2}+\left(-1+x^2\right)\left(-\frac{x^2}{\left(1-x^2\right)^{3/2}}-\frac{1}{\sqrt{1-x^2}}\right)$$

```
Simplify[%]
```

0

So it really works as a solution! But does the Taylor series really give the same as we derived for the series solution?

```
Series[y2, {x,0,6}]
```

$$1-\frac{x^2}{2}-\frac{x^4}{8}-\frac{x^6}{16}+O[x]^7$$

```
OutputForm[%]
```

```
       2    4     6
      x    x     x          7
1 -  -- -  -- -  -- + O[x]
      2    8     16
```

Yes! It compares the same.

What does this mean? It probably means that in creating the problems, Zill, worked backward from a known solution, and created a non-constant coefficient differential equation. Then the solution in terms of series is routine.

Particular Case of Legendre's Equation

Legendre's equation is the differential equation

(1 -x^2) y'' - 2 x y' + n(n+1) y = 0

where n is an integer. I will examine the case when n = 3.

Legendre's equation comes up in many physical situations involving spherical symmetry.

```
y = Sum[ c[i] x^i, { i, 0,6}] + O[x]^7
```

$$c[0] + c[1]\, x + c[2]\, x^2 + c[3]\, x^3 + c[4]\, x^4 + c[5]\, x^5 + c[6]\, x^6 + O[x]^7$$

Insert into the differential equation

```
de = (1 -x^2) D[y, {x,2}] - 2 x D[y,x] + 12 y == 0
```

$$(12\, c[0] + 2\, c[2]) + (10\, c[1] + 6\, c[3])\, x + (6\, c[2] + 12\, c[4])\, x^2 + 20\, c[5]\, x^3 + (-8\, c[4] + 30\, c[6])\, x^4 + O[x]^5 == 0$$

Get the coefficient equations

```
coeffeqns = LogicalExpand[ de]
```

$$12\, c[0] + 2\, c[2] == 0\ \&\&\ 10\, c[1] + 6\, c[3] == 0\ \&\&\ 6\, c[2] + 12\, c[4] == 0\ \&\&\ 20\, c[5] == 0\ \&\&\ -8\, c[4] + 30\, c[6] == 0$$

```
solvedcoeffs = Solve[ coeffeqns, Table[ c[i], {i,2,12}]]
```

Solve::svars : Equations may not give solutions for all "solve" variables. ≫

$$\left\{\left\{c[2] \to -6\, c[0],\ c[3] \to -\frac{5\, c[1]}{3},\ c[4] \to 3\, c[0],\ c[5] \to 0,\ c[6] \to \frac{4\, c[0]}{5}\right\}\right\}$$

Put these back into the series expansion.

```
y = y /. solvedcoeffs
```

$$\left\{c[0]+c[1]\,x-6\,c[0]\,x^2-\frac{5}{3}\,c[1]\,x^3+3\,c[0]\,x^4+\frac{4}{5}\,c[0]\,x^6+O[x]^7\right\}$$

Extract out the two linearly independent solutions

```
Coefficient[ y, c[0]]
```

$$\left\{1-6\,x^2+3\,x^4+\frac{4\,x^6}{5}\right\}$$

```
Coefficient[ y, c[1]]
```

$$\left\{x-\frac{5\,x^3}{3}\right\}$$

This solution is actiually a polynomial, in this case of degree 3. By solving Legendre's equation equation for any integer n, we obtain the full set of corresponding Legendre polynomials of all degrees. The Legendre polynomials are anoother important set of mathematical functions with amazing and wonderful properties. In fact, they are so useful, they are built into Mathematica automatically.

```
LegendreP[3,x]
```

$$\frac{1}{2}\left(-3\,x+5\,x^3\right)$$

This is different from the polynomial obtained earlier, but only by a constant factor. The constant factor is used to set the normailization for the Legendre polynomials.

Exercises

Exercise 1

(1) (a) Find approximate series for the two linearly independent solutions to the equation

$$x''(t) + (1- t/100) x(t) = 0$$

by using the power-series method.

Exercise 2

(2) (a) Find approximate series for the two linearly independent solutions to the equation

$$x''(t) + (1- t/100 + t^2/(2*100^2) x(t) = 0$$

by using the power-series method.

(b) Find the solution with $x(0) =1$ and $x'(0) = 0$.

Don't make this too hard, you should almost be able to do this in your head!